

STAT 5200 Handout #15: Three-Way Mixed Effects Design (Ch. 12) – With Approximate F-Test (Ch. 11)

Example: *Staphylococcus aureus* is the most common of the Staphylococcus family of bacteria, with about 20% of the U.S. population being long-term carriers. Different strains of *Staphylococcus aureus* can cause a variety of ailments from minor skin infections, such as pimples and boils, to life-threatening diseases such as pneumonia, meningitis, toxic shock syndrome, and septicemia. Correct identification of *Staphylococcus aureus* strains is important to ensure that patients are treated appropriately.

At the time of this experiment the protocols for culturing Staphylococcus bacteria were: Time = 24 hours, Temperature = 35° C, and tryptone concentration = 1.0%. A researcher conducted an experiment to find combinations of tryptone concentrations and incubation temperatures that resulted in higher bacterial counts. In the experiment tryptone concentrations of 0.6%, 0.8%, 1.0%, 1.2%, and 1.4% and incubation temperatures of 27 C, 35 C, and 43 C were used. There are many strains of *Staphylococcus aureus*; the experimenter randomly selected 5 strains for this experiment, which are simply coded as 1, 2, 3, 4, and 5.

In a completely random order, one sample of each of the $5 \times 3 \times 5 = 75$ combinations of Staphylococcus aureus strain, temperature, and tryptone concentration is created and after 24 hours a single measurement of the bacterial count in each sample is made, again in a random order.

This is a $5 \times 3 \times 5$ factorial experiment with one observation per combination of factor levels. Temperature and tryptone concentration are fixed factors, but strain is a random factor.

→ unreplicated design (can't keep all interactions)

↳ 5 levels representative of all possible strains

```

/* STAT 5200
   three-way mixed effects design
   -- unreplicated, and approximate F-test
   tryptone data
*/
/* Define options */
ods html image_dpi=300 style=journal;

/* Read in data */
data tryptone; input Temp Tryptone Strain Count @@; cards;
27 0.6 1 15 35 1.2 2 159 43 0.8 4 82
35 0.6 1 80 43 1.2 2 124 27 1.0 4 31
43 0.6 1 53 27 1.4 2 36 35 1.0 4 105
27 0.8 1 31 35 1.4 2 132 43 1.0 4 123
35 0.8 1 82 43 1.4 2 124 27 1.2 4 40
43 0.8 1 77 27 0.6 3 14 35 1.2 4 174
27 1.0 1 32 35 0.6 3 88 43 1.2 4 137
35 1.0 1 99 43 0.6 3 65 27 1.4 4 25
43 1.0 1 117 27 0.8 3 27 35 1.4 4 141
27 1.2 1 40 35 0.8 3 84 43 1.4 4 120
35 1.2 1 179 43 0.8 3 65 27 0.6 5 33
43 1.2 1 138 27 1.0 3 34 35 0.6 5 73
27 1.4 1 32 35 1.0 3 97 43 0.6 5 57

```

```

35 1.4 1 141    43 1.0 3 102    27 0.8 5 17
43 1.4 1 124    27 1.2 3 41    35 0.8 5 96
27 0.6 2 7     35 1.2 3 183    43 0.8 5 83
35 0.6 2 78    43 1.2 3 144    27 1.0 5 39
43 0.6 2 60    27 1.4 3 42    35 1.0 5 100
27 0.8 2 19    35 1.4 3 142    43 1.0 5 121
35 0.8 2 88    43 1.4 3 128    27 1.2 5 35
43 0.8 2 64    27 0.6 4 14    35 1.2 5 178
27 1.0 2 14    35 0.6 4 93    43 1.2 5 147
35 1.0 2 117   43 0.6 4 70    27 1.4 5 37
43 1.0 2 110   27 0.8 4 32    35 1.4 5 140
27 1.2 2 31    35 0.8 4 86    43 1.4 5 124;

```

run;

```

proc mixed data=tryptone method=type3;
  class Strain Temp Tryptone;
  model Count = Temp|Tryptone; (Fixed)
  random Strain Strain*Temp Strain*Tryptone; (random)
  title1 'PROC MIXED with method = type3';
  title2 'to get expected mean sums of squares';

```

run;

PROC MIXED with method = type3 to get expected mean sums of squares				
Source	DF	Sum of Squares	Mean Square	Expected Mean Square
Temp	2	112610	56305	Var(Residual) + 5 Var(Strain*Temp) + Q(Temp,Temp*Tryptone)
Tryptone	4	40628	10157	Var(Residual) + 3 Var(Strain*Tryptone) + Q(Tryptone,Temp*Tryptone)
Temp*Tryptone	8	13047	1630.8567	Var(Residual) + Q(Temp*Tryptone)
Strain	4	589.6800	147.4200	Var(Residual) + 3 Var(Strain*Tryptone) + 5 Var(Strain*Temp) + 15 Var(Strain)
Strain*Temp	8	223.1200	27.8900	Var(Residual) + 5 Var(Strain*Temp)
Strain*Tryptone	16	758.7200	47.4200	Var(Residual) + 3 Var(Strain*Tryptone)
Residual	32	1788.0800	55.8775	Var(Residual)

A

B

AxB

C

AC

BC

E

Source	Error Term <i>(denom. term in F-statistic)</i>	Error DF	F Value	Pr > F
Temp	MS(Strain*Temp)	8	2018.82	<.0001
Tryptone	MS(Strain*Tryptone)	16	214.19	<.0001
Temp*Tryptone	MS(Residual)	32	29.19	<.0001
Strain	MS(Strain*Temp) + MS(Strain*Tryptone) - MS(Residual)	1.1261	7.59	0.2366
Strain*Temp	MS(Residual)	32	0.50	0.8476
Strain*Tryptone	MS(Residual)	32	0.85	0.6266
Residual

Covariance Parameter Estimates	
Cov Parm	Estimate
Strain	8.5325
Strain*Temp	-5.5975
Strain*Tryptone	-2.8192
Residual	55.8775

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Temp	2	8	2018.82	<.0001
Tryptone	4	16	214.19	<.0001
Temp*Tryptone	8	32	29.19	0.0021

```

proc mixed data=tryptone covtest plots=residualpanel;
  class Strain Temp Tryptone;
  model Count = Temp|Tryptone;
  random Strain Strain*Temp Strain*Tryptone;
  lsmeans Temp|Tryptone / pdiff adjust=tukey;
  title1 'PROC MIXED with default REML estimation';
run;

```

PROC MIXED with default REML estimation

Convergence criteria met.

Covariance Parameter Estimates				
Cov Parm	Estimate	Standard Error	Z Value	Pr > Z
Strain	6.5305	6.9773	0.94	0.1746
Strain*Temp	0	.	.	.
Strain*Tryptone	0	.	.	.
Residual	49.4629	9.3476	5.29	<.0001

Type 3 Tests of Fixed Effects				
Effect	Num DF	Den DF	F Value	Pr > F
Temp	2	8	1138.33	<.0001
Tryptone	4	16	205.35	<.0001
Temp*Tryptone	8	32	32.97	<.0001

NOTE: PROC MIXED doesn't do LINES to summarize post-hoc tests. Instead, this table (below) has a header row where " " represents "minus". So the Tryptone0.6 - Tryptone0.8 estimate is -8.8667, with Tukey-adjusted P-value 0.0235.


Difference of Least Squares Means						
Effect	Temp	Tryptone	_Temp	_Tryptone	Estimate	Adj P
Temp	27		35		-88.6800	<.0001
Temp	27		43		-73.6400	<.0001
Temp	35		43		15.0400	0.0002
Tryptone		0.6		0.8	-8.8667	0.0235
...						
Temp*Tryptone	43	1	43	1.2	-23.4000	0.0008
Temp*Tryptone	43	1	43	1.4	-9.4000	0.7174
Temp*Tryptone	43	1.2	43	1.4	14.000	0.1557

[NOTE: Lengthy LSMEANS output shortened here (but can be very useful)]

```

/* Look at Interaction Plots */
proc glimmix data=tryptone;
  class Strain Temp Tryptone;
  model Count = Temp|Tryptone;
  random Strain Strain*Temp Strain*Tryptone;
  lsmeans Temp*Tryptone / plot=mean(sliceby=Temp join);
  lsmeans Temp*Tryptone / plot=mean(sliceby=Tryptone join);
  title 'Interaction Plots';
run;

```



```

/* Get a few things by hand */
data temp;
  pval_C_approx = 1-cdf('F', 7.5863, 4, 1.1261);
  /* F distribution, test stat 7.5863,
     num df 4, denom df (Satterthwaite approx) 1.1261 */
  pval_C_cochran = 1-cdf('F', 2.6995, 7.4728, 23.8530);
proc print data=temp;
  title 'P-value [by-hand] calculations';
run;

```

P-value [by-hand] calculations

Obs	pval_C_approx	pval_C_cochran
1	0.23658	0.030672